SAS Exercises from the 2002 NCHS Data Users Conference, July 15-17, Washington, D.C.

```
/* Exercise A - Reading in NAMCS data */
*Read NAMCS data into SAS;
*Include the following variables-SEX, AGER(recoded age), PATWT, DIAG13D,
                        PHYCODE, PATCODE, REGION, MED1-MED6,
                        DRUGCL1-DRUGCL6, & SPECR;

libname in1 'c:\2002_DUC';
options nocenter noduplex;

data visit;
infile 'c:\2002_DUC\NAM00' missover lrecl=700;
input
sex 11
ager 564  /*recoded age*/
patwt 307-312
nummed 277
diag13D $ 154-156
phycode $319-322
patcode $323-325
med1 247-251 med2 252-256
med3 257-261 med4 262-266
med5 267-271 med6 272-276
drugcl1 339-342 drugcl2 376-379
drugcl3 413-416 drugcl4 450-453
drugcl5 487-490
drugcl6 524-527

/*sampling design variables*/
STRATM 621-623
PSUM 624-628
YEAR 629-632
SUBFILE 633
PROSTRAT 634-635
PROVIDER 636-639
DEPT 640
SU 641-643
CLINIC 644-646
POPPSUM 647-648
POPSU 649-651
POPVISM 652-657
POPPROVM 658-663;

*Create a new visit weight that is PATWT/1000;

wait=patwt/1000; /*number of visits in 1000's*/
run;

proc contents; run;
proc print data=visit (obs=20);
run;
```

```
/* Exercise B-Getting simple Frequencies */

*Estimate the number of office visits made by females less than 15 years of age.;
proc freq data=visit;
tables sex*ager;
title 'Unweighted 2000 NAMCS';
run;

proc freq data=visit;
tables sex*ager;
weight wait;
title 'Weighted 2000 NAMCS (in thousands)';
run;


*Estimate the number of asthma visits made by females less than 15 years of age.;
proc freq data=visit (where=(DIAG13D='493'));
tables sex*ager;
title 'Unweighted 2000 asthma visits NAMCS';
run;

proc freq data=visit (where=(DIAG13D='493'));
tables sex*ager;
weight wait;
title 'Weighted 2000 NAMCS asthma visits (in thousands)';
run;
```

```
/* Exercise C-Transforming data */
*Do a DATA step to create a variable ASTHMA where 1 is diag13d='493'
                                        and 0 if not;

data visit;
set visit;
if DIAG13D='493' then asthma=1;
              else asthma=0;

*What is the age distribution for asthma visits?;
proc freq data=visit (where=(asthma=1));
tables ager;
title 'Age distribution for asthma cases,unweighted (in thousands)';
run;

proc freq data=visit (where=(asthma=1));
tables ager;
weight wait;
title 'Age distribution for asthma cases, weighted(in thousands)';
run;
```

```
/* Exercise D - Creating a drug file */
*Do a data step where a new record is output every time a MEDCODE
 is NOT 90000;

data drug;
set visit;
array medcod(6) med1-med6;
array clas(6) drugcl1-drugcl6;
   do i=1 to 6;
      if medcod(i) ne 90000 then do; /*90000=blanks*/
        medcode=medcod(i);
        class=clas(i);
      output;
      end;
   end;

proc print data=drug (obs=10);
title '2000 NAMCS drug file';
run;

*What are the top 5 MEDCODEs for asthma visits?;
*Which therapeutic class has the largest number of drug mentions
 for asthma visits?;
proc freq data=drug (where=(asthma=1)) order=freq ;
tables medcode class;
weight wait;
title 'Asthma drug mentions';
run;
```

```
/* Exercise E - Using sample design variables in SUDAAN */

*Proc crosstabs do not like 0's so all categorical variables with 0's must
 be recoded.;

*Data must be sorted by these variables in order to
  run SUDAAN software;

data sud;
set visit;
keep stratm psum year subfile prostrat
provider dept su clinic poppsum popprovm
popsu popvism sex ager wait;

proc sort data=sud;
by stratm psum year subfile prostrat provider dept su clinic;
run;

*Repeat of exercise B-Estimate the number of office visits
made by females less than 15 years of age.;

PROC CROSSTAB data=sud DESIGN = WOR notot;
NEST STRATM PSUM SUBFILE PROSTRAT YEAR PROVIDER DEPT SU CLINIC/MISSUNIT;

TOTCNT POPPSUM _ZERO_ _ZERO_ _ZERO_ POPPROVM _ZERO_ POPSU _ZERO_ POPVISM;
WEIGHT WAIT;
SUBGROUP sex ager;
LEVELS   2 6 ;
TABLES  sex*ager;
SETENV COLWIDTH = 15 decwidth=4;
RTITLE "Repeat of Exercise B-number of visits made by females less than 15";
PRINT/style=nchs;
RUN;
```

```
/* Sample Program for Running 2000 NAMCS Public-Use File with SAS Documentation Downloaded
from Ambulatory Health Care Data website. This documentation provides you with an input statement,
value and variable labels, and format assignments. It is assumed that you have already downloaded the
data files and documentation and saved them in a directory of your choice.  We used c:\2002_DUC for
the purpose of the workshop. */

filename nam00 'c:\2002_DUC\nam00';

filename nam00sas 'c:\2002_DUC\nam00sas.txt';
filename nam00for 'c:\2002_DUC\nam00for.txt';
filename nam00lab 'c:\2002_DUC\nam00lab.txt';


libname out1 'c:\2002_DUC';

%inc nam00for;

data namtest1;
infile nam00 missover lrecl=999;
%inc nam00sas;
%inc nam00lab;

patwt2=patwt/1000;

Proc freq data=namtest1;
weight patwt2;
tables vmonth--popprovm/list;
title '2000 NAMCS Output File';
run;
```

```
/* This program uses downloadable SAS documentation from the Ambulatory Health Care Data website
to read in files from the 2000 NAMCS, 2000 NHAMCS - Outpatient Department File, and 2000 NHAMCS
- Emergency Department File to create a combined file.  A variable is also created to identify each of the
three ambulatory care settings.  It is assumed that you have already downloaded the data files and
documentation and saved them in a directory of your choice.  We used c:\2002_DUC for the purpose
of the workshop. */


filename nam00 'c:\2002_DUC\nam00';
filename nam00sas 'c:\2002_DUC\nam00sas.txt';
filename nam00for 'c:\2002_DUC\nam00for.txt';
filename nam00lab 'c:\2002_DUC\nam00lab.txt';

filename ed00 'c:\2002_DUC\ed00';
filename ed00sas 'c:\2002_DUC\ed00sas.txt';
filename ed00for 'c:\2002_DUC\ed00for.txt';
filename ed00lab 'c:\2002_DUC\ed00lab.txt';

filename opd00 'c:\2002_DUC\opd00';
filename opd00sas 'c:\2002_DUC\opd00sas.txt';
filename opd00for 'c:\2002_DUC\opd00for.txt';
filename opd00lab 'c:\2002_DUC\opd00lab.txt';

libname out1 'c:\2002_DUC';

%inc nam00for;
%inc opd00for;
%inc ed00for;

proc format;
  value setfmt  1='Physician office'
                2='Outpatient department'
                3='Emergency department';

data namtest1;
infile nam00 missover lrecl=999;
%inc nam00sas;
%inc nam00lab;

data edtest1;
infile ed00 missover lrecl=999;
%inc ed00sas;
%inc ed00lab;

data opdtest1;
infile opd00 missover lrecl=999;
%inc opd00sas;
%inc opd00lab;

data alltest1; set namtest1 edtest1 opdtest1;

if subfile=1 then settype=1;
else if subfile=2 and dept=1 then settype=3;
else if subfile=2 and dept=2 then settype=2;

patwt2=patwt/1000;
```

```
proc freq data=alltest1;
weight patwt2;
tables paytype/list;
format settype setfmt.;
run;

Proc freq data=alltest1;
weight patwt2;
tables paytype*settype/list;
format settype setfmt.;

run;
```